SET INTERSECTION

PURPOSE
Carry out the intersection of 2 sets with numeric elements.

DESCRIPTION
The intersection of 2 sets is the set containing the elements common to both sets. For example, the intersection of the 5-element set 1 3 5 7 9 and the 4-element set 1 4 9 16 is the 2-element set 1 9.

SYNTAX
LET <v3> = SET INTERSECTION <v1> <v2> <SUBSET/EXCEPT/FOR qualification>
where <v1> is the variable containing the elements of the first set;
<v2> is the variable containing the elements of the second set;
<v3> is the variable containing the elements of the resultant set;
and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

EXAMPLES
LET Y3 = SET INTERSECTION Y1 Y2
LET Y3 = SET INTERSECTION Y1 Y2 SUBSET Y1 > 10

NOTE
If the elements of a mathematical “set” are numbers (or can be translated into numbers-- always possible), then a DATAPLOT variable can be used to store the items of the mathematical set. To store the set with the 12 elements 1 3 5 7 11 1 4 9 16 18 27, form the variable Y with the following command:
LET Y = DATA 1 3 5 7 11 1 4 9 16 18 27

Larger sets can be created with the READ or SERIAL READ commands.

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
SET CARDINALITY = Computes the number of elements in a set.
SET UNION = Carries out a set union.
SET COMPLEMENT = Carries out a set complement.
SET CARTESIAN PRODUCT = Carries out a set Cartesian product.

APPLICATIONS
Mathematics

IMPLEMENTATION DATE
87/10
PROGRAM

. PURPOSE--DETERMINE THE SOLUTIONS OF A DIOPHANTINE EQUATION--
. FIND THE INTEGER SOLUTIONS (X AND Y) OF X**2 + 1 = 2*Y**4
. ANALYSIS TECHNIQUE--EVALUATION AND SET INTERSECTION WITH INTEGERS
. SOURCE--STEEN, LYNN ARTHUR, MATHEMATICS TODAY,
. VINTAGE BOOKS, NEW YORK, 1980, PAGE 40.
. APPLICATION--

. -----START POINT-----------------------------------

. STEP 1--DEFINE THE FUNCTION OF INTEREST IN Y = F(X) FORM.
. DEFINE A SEQUENCE OF X VALUES AND COMPUTE CORRESPONDING Y VALUES.
LET FUNCTION F = ((X**2+1)/2)**0.25
LET X = SEQUENCE 0 1 500
LET Y = F

. STEP 2--DETERMINE THE INTERSECTION OF THE Y VALUES
. WITH THE (INTEGER) X VALUES.
LET Y2 = SET INTERSECTION X Y
PRINT Y2

. STEP 3--IN ANOTHER FASHION, DETERMINE THE Y VALUES WHICH ARE
. INTEGER AND THE (NECESSARILY INTEGER) X VALUES
LET Y3 = FRACT(Y)
LINE SOLID BLANK
CHARACTER BLANK X
PLOT Y X AND
PLOT Y X SUBSET Y3 0